

Handout 4

PERBANDINGAN DUA VEKTOR NILAI TENGAH

1. Uji Hipotesis

a. Sampel Berpasangan

$$H_0 : \mathbf{d} = \boldsymbol{\mu}_1 - \boldsymbol{\mu}_2 = \mathbf{0}$$

$$H_1 : \mathbf{d} = \boldsymbol{\mu}_1 - \boldsymbol{\mu}_2 \neq \mathbf{0}$$

- Statistik Uji :

$$T^2 = n\mathbf{d}'\mathbf{S}^{-1}\mathbf{d}$$

- Tolak H_0 jika $T^2 = n\mathbf{d}'\mathbf{S}^{-1}\mathbf{d} > c^2 = \frac{(n-1)p}{(n-p)} F_{(p,n-p)}(\alpha)$

dengan n adalah banyaknya sampel dan p adalah banyaknya peubah (*variable*)

b. Sampel Saling Bebas dengan $\boldsymbol{\Sigma}_1 = \boldsymbol{\Sigma}_2$

$$H_0 : \boldsymbol{\mu}_1 = \boldsymbol{\mu}_2$$

$$H_1 : \boldsymbol{\mu}_1 \neq \boldsymbol{\mu}_2$$

- Statistik Uji :

$$T^2 = n(\bar{\mathbf{x}}_1 - \bar{\mathbf{x}}_2)' \left[\left(\frac{1}{n_1} + \frac{1}{n_2} \right) \mathbf{S}_{gab} \right]^{-1} (\bar{\mathbf{x}}_1 - \bar{\mathbf{x}}_2)$$

Dengan

$$\mathbf{S}_{gab} = \frac{(n_1 - 1)\mathbf{S}_1 + (n_2 - 1)\mathbf{S}_2}{n_1 + n_2 - 2}$$

- Tolak H_0 jika $T^2 > c^2 = \frac{(n_1+n_2-2)p}{n_1+n_2-p-1} F_{(p,n_1+n_2-p-1)}(\alpha)$

dengan n adalah banyaknya sampel dan p adalah banyaknya peubah (*variable*)

c. Sampel Saling Bebas dengan $\boldsymbol{\Sigma}_1 \neq \boldsymbol{\Sigma}_2$

$$H_0 : \boldsymbol{\mu}_1 = \boldsymbol{\mu}_2$$

$$H_1 : \boldsymbol{\mu}_1 \neq \boldsymbol{\mu}_2$$

- Statistik Uji :

$$T^2 = n(\bar{\mathbf{x}}_1 - \bar{\mathbf{x}}_2)' \left(\frac{\mathbf{S}_1}{n_1} + \frac{\mathbf{S}_2}{n_2} \right)^{-1} (\bar{\mathbf{x}}_1 - \bar{\mathbf{x}}_2)$$

- Tolak H_0 jika $T^2 > \chi^2_{(\alpha,p)}$

2. Selang Kepercayaan Ellips $(1 - \alpha)100\%$ bagi μ

a. untuk sampel berpasangan

$$n(\bar{d} - \delta)' S_d^{-1} (\bar{d} - \delta) \leq c^2$$

b. Untuk sampel yang saling bebas

$$n(\bar{x} - \mu)' S^{-1} (\bar{x} - \mu) \leq c^2$$

kok X dan miu? Bukannya selisih ya?

3. Selang Kepercayaan Simultan $(1 - \alpha)100\%$ bagi μ

a. Untuk sampel berpasangan

$$\delta_i : \bar{d}_i \pm c \sqrt{\frac{S_{d_i}}{n}}$$

b. Untuk sampel yang saling bebas

- $\Sigma_1 = \Sigma_2$

$$a(\mu_1 - \mu_2) : a'(\bar{x}_1 - \bar{x}_2) \pm c \sqrt{a' \left(\frac{1}{n_1} + \frac{1}{n_2} \right) S_{gab} a}$$

- $\Sigma_1 \neq \Sigma_2$

$$a(\mu_1 - \mu_2) : a'(\bar{x}_1 - \bar{x}_2) \pm c \sqrt{\chi^2_{(\alpha, p)} a' \left(\frac{S_1}{n_1} + \frac{S_2}{n_2} \right) a}$$

4. Selang Kepercayaan Bonferroni $(1 - \alpha)100\%$ bagi μ

a. Untuk sampel berpasangan

$$\delta_i : \bar{x}_i \pm t_{(n-1)} \left(\frac{\alpha}{2p} \right) \sqrt{\frac{S_{d_i}}{n}}$$

b. Untuk sampel yang saling bebas

- $\Sigma_1 = \Sigma_2$

$$(\mu_{1i} - \mu_{2i}) : (\bar{x}_{1i} - \bar{x}_{2i}) \pm t_{(n_1+n_2-1)} \left(\frac{\alpha}{2p} \right) \sqrt{\left(\frac{1}{n_1} + \frac{1}{n_2} \right) S_{ii}}$$

Contoh Kasus:

1. Seorang guru memberikan tugas kepada 15 orang siswanya untuk menulis dua jenis essay, yaitu essay yang bersifat formal dan yang bersifat informal. Penilaian baik buruknya essay dilihat dari dua aspek, yaitu banyaknya kata dan banyaknya kata kerja yang termuat di dalam essay. Semakin banyak kata dan kata kerja yang termuat di essay, maka semakin bagus penilaiannya. Berikut adalah data hasil penulisan essay yang telah dibuat oleh 15 siswa tersebut.

Siswa	Essay Informal		Essay Formal	
	Kata	Kata Kerja	Kata	Kata Kerja
1	148	20	137	15
2	159	24	164	25
3	144	19	224	27
4	103	18	208	33
5	121	17	178	24
6	89	11	128	20
7	119	17	154	18
8	123	13	158	16
9	76	16	102	21
10	217	29	214	25
11	148	22	209	24
12	151	21	151	16
13	83	7	123	13
14	135	20	161	22
15	178	15	175	23

- a. Lakukan pengujian, apakah essay informal buatan siswa-siswa tersebut lebih bagus hasilnya daripada essay formal? Gunakan $\alpha = 0.05$.
- b. Buatlah selang kepercayaan simultan dan selang bonferoni 95% bagi selisih nilai tengah δ_i

PEMBAHASAN :

a. Hipotesis:

$$H_0: \delta = 0$$

$$H_1: \delta \neq 0$$

Statistik Uji:

$$T^2 = n(\bar{\mathbf{d}} - \boldsymbol{\delta})' \mathbf{S}_d^{-1} (\bar{\mathbf{d}} - \boldsymbol{\delta})$$

dimana;

$$n = 15$$

$$\mathbf{d}_j = \mathbf{x}_{1j} - \mathbf{x}_{2j} \quad ; j=1,2, \dots 15 \text{ (pasangan selisih peubah acak)}$$

$$\bar{\mathbf{d}} = \frac{1}{n} \sum_{j=1}^n \mathbf{d}_j$$

$$\mathbf{S}_d = \frac{1}{n-1} \sum_{j=1}^n (\mathbf{d}_j - \bar{\mathbf{d}}) (\mathbf{d}_j - \bar{\mathbf{d}})'$$

Dari perhitungan diperoleh hasil sebagai berikut;

$$\bar{\mathbf{d}} = \begin{pmatrix} 32,8 \\ 3,5 \end{pmatrix}$$

$$\mathbf{S}_d = \begin{pmatrix} 1096,028 & 139,9 \\ 139,9 & 31,552 \end{pmatrix}$$

$$\mathbf{S}_d^{-1} = \begin{pmatrix} 0,0021 & -0,0093 \\ -0,0093 & 0,073 \end{pmatrix}$$

Sehingga diperoleh statistik uji T^2 :

$$\begin{aligned} T^2 &= 15 \begin{pmatrix} 32,8 & 3,5 \end{pmatrix}' \begin{pmatrix} 0,0021 & -0,0093 \\ -0,0093 & 0,073 \end{pmatrix} \begin{pmatrix} 32,8 \\ 3,5 \end{pmatrix} \\ &= 15,19123 \end{aligned}$$

Titik kritis:

$$\frac{(n-1)p}{(n-p)} F_{p,n-p}(\alpha) = 8,1966$$

Keputusan:

statistik uji $T^2 >$ titik kritis maka cukup bukti untuk menolak H_0 .

Kesimpulan:

Dengan tingkat kepercayaan 95%, dapat disimpulkan bahwa nilai essay informal buatan siswa-siswa tersebut berbeda hasilnya daripada essay formal. Berdasarkan rata-rata dapat dikatakan bahwa nilai essay formal buatan siswa-siswa tersebut lebih bagus hasilnya daripada essay informal.

Syntax Program SAS :

```
/*PAIRED HOTELLING TEST*/
```

```
data essay;
input case iwords iverbs fwords fverbs;
dwords = fwords-iwords;
dverbs = fverbs-iverbs;
label iwords='Number of words in informal essay'
      iverbs='Number of verbs in informal essay'
      fwords='Number of words in formal essay'
      fverbs='Number of verbs in formal essay'
;
datalines;
1 148 20 137 15
2 159 24 164 25
3 144 19 224 27
4 103 18 208 33
5 121 17 178 24
6 89 11 128 20
7 119 17 154 18
8 123 13 158 16
9 76 16 102 21
10 217 29 214 25
11 148 22 209 24
12 151 21 151 16
13 83 7 123 13
14 135 20 161 22
15 178 15 175 23
;
proc means mean var;
var fwords fverbs iwords iverbs;
title 'Check whether mean=variance';
run;

proc iml;
start samplestats(X,Xbar,W,S,R,n);
n = nrow(X);
one = J(n,1);
Xbar = X`*one/n;
W = (X - one*Xbar)` * (X - one*Xbar);
S = W/(n-1);
Dsqr = sqrt(diag(S));
R = inv(Dsqr)*S*inv(Dsqr);
Finish samplestats;

use essay;
read all var{fwords fverbs} into Xf;
read all var{iwords iverbs} into Xi;

d = Xf-Xi;
run samplestats(d,dbar,Wd,Sd,Rd,nd);

T2 = nd*dbar`*inv(Sd)*dbar;

p= ncol(Xf);
dfden=nd-p;

Fcrit = quantile('F',.95, p,dfden) ;
```

```

Fcrit = (nd-1)*p/(nd-p) * Fcrit;
F = (nd-p)/((nd-1)*p)*T2;
pvalue = 1- cdf('F',F,p,dfden);

print 'Results for the paired Comparison data on essays', ,
      'Sample Size =', nd, ,
      'Sample means =', dbar, ,
      'Sample Covariance =', Sd, ,
      'Hotellings T^2 =', T2, ,
      '(nd-1)*p/(nd-p) * F_p,n-p(.05) =', Fcrit, ,
      '(nd-p)/((nd-1)*p) T^2 =', F, ,
      'p-value for (nd-p)/((nd-1)*p) T^2=' pvalue ,;
call eigen(lambda,U,Sd);
print lambda U;
major1 = dbar + sqrt(lambda[1,1])*sqrt(Fcrit/nd)*U[,1];
major2 = dbar - sqrt(lambda[1,1])*sqrt(Fcrit/nd)*U[,1];
minor1 = dbar + sqrt(lambda[2,1])*sqrt(Fcrit/nd)*U[,2];
minor2 = dbar - sqrt(lambda[2,1])*sqrt(Fcrit/nd)*U[,2];
print 'Points of Confidence Region:' major1 major2 minor1 minor2;

```

Output :

Check whether mean=variance

10

13:00 Thursday, October 14, 2014

The MEANS Procedure

Variable	Label	Mean	Variance
~~~~~			
fwords	Number of words in formal essay	165.7333333	1299.78
fverbs	Number of verbs in formal essay	21.4666667	27.9809524
iwords	Number of words in informal essay	132.9333333	1405.78
iverbs	Number of verbs in informal essay	17.9333333	28.6380952
~~~~~			

Check whether mean=variance

11

13:00 Thursday, October 14, 2014

Results for the paired Comparison data on essays

ND

Sample Size =

15

DBAR

Sample means =

32.8
3.5333333

SD

Sample Covariance =

1096.0286 139.9
139.9 31.552381

T2

Hotellings T^2 =

15.191234

FCRIT

(nd-1)*p/(nd-p) * F_p,n-p(.05) = 8.1966021

F

(nd-p)/((nd-1)*p) T^2 =

7.0530731

PVALUE

p-value for $(nd-p)/((nd-1)*p)$ $T^2 = 0.0084274$

LAMBDA U

1114.108 0.9917528 -0.128165
13.472931 0.1281654 0.9917528

MAJOR1 MAJOR2 MINOR1 MINOR2

Points of Confidence Region: 57.270249 8.3297513 32.452245 33.147755
6.6956537 0.3710129 6.2242843 0.8423824

Dari Output SAS di atas diperoleh T^2 Hotellings 15.191234, dan p-value $(0.0084274) < \alpha (0,05)$ sehingga tolak H_0 yang berarti bahwa ada perbedaan antara hasil essay formal dan informal buatan siswa saat dilakukan pengujian dengan taraf nyata 5%. Jika dilihat dari rata-rata keseluruhan, maka benar bahwa essay formal lebih bagus hasilnya daripada essay informal.

b. Selang Kepercayaan

- Silmultan

Selang kepercayaan simultan bagi selisih nilai tengah δ_i :

$$\bar{d}_i - \sqrt{\frac{(n-1)p}{(n-p)} F_{p,n-p}(\alpha)} \sqrt{\frac{S_{d_i}^2}{n_i}} < \delta_i < \bar{d}_i + \sqrt{\frac{(n-1)p}{(n-p)} F_{p,n-p}(\alpha)} \sqrt{\frac{S_{d_i}^2}{n_i}}$$

Dimana, \bar{d}_i adalah elemen ke-i dari \bar{d} dan $S_{d_i}^2$ adalah elemen pada diagonal ke-i dari S_d .

Sehingga;

selang kepercayaan simultan bagi δ_1 :

$$32,8 - \sqrt{8,1966} \sqrt{\frac{1096,028}{15}} < \delta_1 < 32,8 + \sqrt{8,1966} \sqrt{\frac{1096,028}{15}}$$

$$8,3272 < \delta_1 < 57,2727$$

selang kepercayaan simultan bagi δ_2 :

$$3,5 - \sqrt{8,1966} \sqrt{\frac{31,552}{15}} < \delta_2 < 3,5 + \sqrt{8,1966} \sqrt{\frac{31,552}{15}}$$

$$-0,6189 < \delta_2 < 7,685$$

Syntak Program SAS :

```
***** Simultaneous component T2 intervals *****/  
d_lower = dbar - sqrt( Fcrit*vecdiag(Sd/nd) );  
d_upper = dbar + sqrt( Fcrit*vecdiag(Sd/nd) );  
print 'Simultaneous T2 Intervals: ' d_lower d_upper;
```

Output :

	D_LOWER	D_UPPER
Simultaneous T2 Intervals:	8.3272804	57.27272
	-0.618954	7.6856203

Dari output SAS diketahui bahwa selang kepercayaan simultan 95% dari selisih nilai kata informal terhadap formal (informal-formal) berada pada selang 8,3272 sampai 57,2727, dan selang kepercayaan simultan 95% dari selisih nilai kata kerja informal terhadap formal (informal-formal) yang berada pada selang -0,61893 sampai 7,68559.

- Bonferroni

Selang kepercayaan 100 (1- α)% bonferroni untuk selisih nilai tengah δ_i adalah:

$$\bar{d}_i - t_{n-1} \left(\frac{\alpha}{2p} \right) \sqrt{\frac{S_{d_i}^2}{n_i}} < \delta_i < \bar{d}_i + t_{n-1} \left(\frac{\alpha}{2p} \right) \sqrt{\frac{S_{d_i}^2}{n_i}}$$

Dimana $t_{n-1} \left(\frac{\alpha}{2p} \right)$ adalah persentil ke-100($\alpha/2p$) dari sebaran t dengan $db = n-1$

Sehingga;

selang kepercayaan bonferroni bagi δ_1 :

$$32,8 - 2,5095 \sqrt{\frac{1096,028}{15}} < \delta_1 < 32,8 + 2,5095 \sqrt{\frac{1096,028}{15}}$$
$$11,3481 < \delta_1 < 54,2518$$

selang kepercayaan bonferroni bagi δ_2 :

$$3,5 - 2,5095 \sqrt{\frac{31,552}{15}} < \delta_2 < 3,5 + 2,5095 \sqrt{\frac{31,552}{15}}$$
$$-0,106 < \delta_2 < 7,173$$

Syntax Program SAS :


```

/***** Bonferroni intervals*****/
alpha_2m = 1-.05/(2*2);
df = nd -1;
tstat = quantile('t',alpha_2m,df);
db_lower = dbar - tstat* sqrt(vecdiag(Sd/nd));
db_upper = dbar + tstat* sqrt(vecdiag(Sd/nd));
print 'Bonferroni Intervals:      ' db_lower db_upper;

```

Output :

	DB_LOWER	DB_UPPER
Bonferroni Intervals:	11.348161	54.251839
	-0.106401	7.1730674

Dari output di atas diperoleh selang kepercayaan Bonferroni 95% dari selisih nilai kata informal terhadap formal (informal-formal) yang berada pada selang 11,348161 sampai 54,251839. Serta diperoleh selang kepercayaan Bonferroni 95% dari selisih nilai kata kerja informal terhadap formal (informal-formal) yang berada pada selang -0,106401 sampai 7,1730674

2. Di Indonesia sejak tahun 2013 diberlakukan proses seleksi Calon Pegawai Negeri Sipil (CPNS) menggunakan sistem Computer Assited Test (CAT). Namun, pada tahun 2013, belum seluruh instansi penerima CPNS menggunakan sistem ini. Materi soal yang diujikan pada tes tahap pertama (Tes Kompetensi Dasar/TKD) terdiri dari tiga bagian, yaitu Tes Wawasan Kebangsaan (TWK), Tes Intelegensi Umum (TIU) dan Tes Karakteristik Pribadi (TKP). Dari ketiga bagian soal tersebut, yang membutuhkan kemampuan kognitif yang jawabannya mutlak benar atau salah adalah TWK dan TIU, sedangkan TKP mengukur kematangan mental seseorang dan kesesuaiannya dengan pekerjaan yang dilamar. Ada dugaan bahwa rata-rata nilai hasil test untuk TWK dan TIU dari peserta yang mengikuti seleksi menggunakan CAT berbeda dengan peserta yang mengikuti seleksi menggunakan sistem manual (lembar jawaban komputer/LJK). Untuk membuktikan kebenaran dugaan itu, sesorang melakukan survey dengan mengambil sampel 20 orang peserta yang telah mengikuti test menggunakan CAT dan 20 orang yang mengikuti test secara manual. Hasil nilai yang diperoleh adalah sebagai berikut :

NO	MANUAL		CAT	
	TWK	TIU	TWK	TIU
1	58	84	54	77
2	49	60	76	58
3	83	69	55	58
4	55	60	79	63
5	85	89	84	37
6	90	68	60	45
7	42	36	73	60
8	85	57	81	56
9	80	46	65	80
10	89	51	54	42
11	40	65	67	42
12	71	64	51	64
13	81	82	56	62
14	40	50	50	41
15	80	77	58	43
16	75	35	82	46
17	41	55	61	48
18	54	61	60	43
19	48	50	45	56

20	69	43	80	38
----	----	----	----	----

- Lakukan pengujian menggunakan taraf nyata 5% apakah dugaan tersebut benar?
- Tentukan selang kepercayaan simultan dan bonferoni 95%.

PEMBAHASAN:

KASUS RAGAM SAMA

a. Hipotesis :

$$H_0: \mu_1 = \mu_2$$

$$H_1: \mu_1 \neq \mu_2$$

Statistik uji :

$$T^2 = (\bar{\mathbf{x}}_1 - \bar{\mathbf{x}}_2)' \left[\left(\frac{1}{n_1} + \frac{1}{n_2} \right) S_{gab} \right]^{-1} (\bar{\mathbf{x}}_1 - \bar{\mathbf{x}}_2) > c^2$$

Dimana:

$$S_{gab} = \frac{(n_1 - 1)S_1 + (n_2 - 1)S_2}{n_1 + n_2 - 2}$$

$$c^2 = \frac{(n_1 + n_2 - 2)p}{(n_1 + n_2 - p - 1)} F_{p, n_1 + n_2 - p - 1}(\alpha)$$

Matriks kovarians:

$$S_1 = \begin{bmatrix} \text{var}(x_1) & \text{covar}(x_1 x_2) \\ \text{covar}(x_2 x_1) & \text{var}(x_2) \end{bmatrix} = \begin{bmatrix} 333.776 & 81.868 \\ 81.868 & 230.441 \end{bmatrix}$$

$$S_2 = \begin{bmatrix} \text{var}(x_1) & \text{covar}(x_1 x_2) \\ \text{covar}(x_2 x_1) & \text{var}(x_2) \end{bmatrix} = \begin{bmatrix} 151.103 & -26.287 \\ -26.287 & 154.155 \end{bmatrix}$$

Matriks kovarians gabungan:

$$S_{gab} = \frac{(n_1 - 1)S_1 + (n_2 - 1)S_2}{n_1 + n_2 - 2} = \begin{bmatrix} 242.4495 & 27.7905 \\ 27.7905 & 192.298 \end{bmatrix}$$

$$\left(\frac{1}{n_1} + \frac{1}{n_2} \right) S_{gab} = \begin{bmatrix} 24,24395 & 2,77905 \\ 2,77905 & 19,2283 \end{bmatrix}$$

$$\left[\left(\frac{1}{n_1} + \frac{1}{n_2} \right) S_{gab} \right]^{-1} = \begin{bmatrix} 0,041942 & -0,00606 \\ -0,00606 & 0,052883 \end{bmatrix}$$

Statistik uji T^2 :

$$T^2 = (\bar{\mathbf{x}}_1 - \bar{\mathbf{x}}_2)' \left[\left(\frac{1}{n_1} + \frac{1}{n_2} \right) S_{gab} \right]^{-1} (\bar{\mathbf{x}}_1 - \bar{\mathbf{x}}_2)$$

$$T^2 = [1,2 \quad 7,15] \begin{bmatrix} 0,041942 & -0,00606 \\ -0,00606 & 0,052883 \end{bmatrix}^{-1} \begin{bmatrix} 1,2 \\ 7,15 \end{bmatrix} = \mathbf{2,6598}$$

Nilai titik kritis :

$$c^2 = \frac{(n_1 + n_2 - 2)p}{(n_1 + n_2 - p - 1)} F_{p, n_1 + n_2 - p - 1}(\alpha) = \frac{38 \times 2}{37} \times 3.25194 = 6.6798$$

Kesimpulan:

Pada Saat ragam Populasi Tes CPNS Manual diasumsikan Sama dengan Ragam Populasi Tes CPNS CAT, Maka dapat disimpulkan belum cukup bukti menolak H_0 . Dengan kata lain, nilai Tes CPNS manual tidak berbeda dengan tes CPNS Cat dengan uji taraf nyata 5%.

Dengan menggunakan program SAS, berikut ini adalah syntak yang digunakan :

```
Data CPNS;
input metode$ twk tiu;
datalines;
Manual 58      84
Manual 49      60
Manual 83      69
Manual 55      60
Manual 85      89
Manual 90      68
Manual 42      36
Manual 85      57
Manual 80      46
Manual 89      51
Manual 40      65
Manual 71      64
Manual 81      82
Manual 40      50
Manual 80      77
Manual 75      35
Manual 41      55
Manual 54      61
Manual 48      50
Manual 69      43
cat      54      77
cat      76      58
cat      55      58
cat      79      63
cat      84      37
cat      60      45
cat      73      60
cat      81      56
```

cat	65	80
cat	54	42
cat	67	42
cat	51	64
cat	56	62
cat	50	41
cat	58	43
cat	82	46
cat	61	48
cat	60	43
cat	45	56
cat	80	38

```

;
proc print data=CPNS;
run;

/* uji 2 sampel ragam sama*/

proc iml;
  start hotel2;
    n1=nrow(x1);
    n2=nrow(x2);
    k=ncol(x1);
    one1=j(n1,1,1);
    one2=j(n2,1,1);
    ident1=i(n1);
    ident2=i(n2);
    ybar1=x1`*one1/n1;
    s1=x1`*(ident1-one1*one1`/n1)*x1/(n1-1.0);
    ybar2=x2`*one2/n2;
    s2=x2`*(ident2-one2*one2`/n2)*x2/(n2-1.0);
    spool=((n1-1.0)*s1+(n2-1.0)*s2)/(n1+n2-2.0);
    t2=(ybar1-ybar2)`*inv(spool*(1/n1+1/n2))*(ybar1-ybar2);
    f=(n1+n2-k-1)*t2/k/(n1+n2-2);
    df1=k;
    df2=n1+n2-k-1;
    p=1-probf(f,df1,df2);
    print t2 f df1 df2 p;
  finish;
  use CPNS;
  read all var{twk tiu} where (metode="Manual") into x1;
  read all var{twk tiu} where (metode="cat") into x2;
  run hotel2;

```

Output yang diperoleh :

The SAS System				
T2	F	DF1	DF2	P
2.659877	1.2949401	2	37	0.2860384

P-value pada output di atas adalah $0,2860 > \alpha(0,05)$ yang sehingga tidak cukup bukti untuk menolak H_0 yang berarti bahwa rata-rata nilai hasil test untuk TWK dan TIU dari peserta yang mengikuti seleksi menggunakan CAT tidak berbeda dengan peserta yang mengikuti seleksi menggunakan sistem manual (lembar jawaban komputer/LJK) melalui pengujian dengan taraf nyata 5%.

b. Tentukan selang kepercayaan 95%.

- Selang Kepercayaan Simultan

$$a' \mu \leq a' (\bar{x}_1 - \bar{x}_2) \pm \sqrt{\frac{(n_1 + n_2 - 2)p}{(n_1 + n_2 - p - 1)} F_{p, n-p}(\alpha)} \sqrt{a' \left(\frac{1}{n_1} + \frac{1}{n_1} \right) s_{gab} a}$$

Selang kepercayaan 95% $\mu_1 - \mu_2$ bagi TWK:

$$-11.5255 \leq \mu_1 \leq 13.9255$$

Interpretasi :

dengan tingkat kepercayaan 95% selisih nilai TWK tes manual dan CAT adalah antara -11.5255 hingga 13.9255. Selang kepercayaan ini memuat nilai 0 didalamnya sehingga dapat disimpulkan bahwa TWK antara manual dan CAT tidak berbeda nyata.

Selang kepercayaan 95% $\mu_1 - \mu_2$ bagi TIU:

$$-4.183 \leq \mu_2 \leq 18.483$$

Interpretasi:

dengan tingkat kepercayaan 95% selisih nilai TIU tes manual dan CAT adalah antara -4.183 hingga 18.434. Selang kepercayaan ini memuat nilai 0 didalamnya sehingga dapat disimpulkan bahwa TIU antara manual dan CAT tidak berbeda nyata.

- Selang kepercayaan bonferroni

$$(\bar{x}_1 - \bar{x}_2) \pm t_{(n_1+n_2-2), \left(\frac{\alpha}{2p}\right)} \sqrt{\left(\frac{1}{n_1} + \frac{1}{n_1}\right) S_{ii}}$$

p = banyaknya selang kepercayaan yang akan di buat.

Selang kepercayaan 95% $\mu_1 - \mu_2$ bagi TWK:

$$-7.1013 \leq \mu_1 \leq 9.5013$$

Interpretasi:

dengan tingkat kepercayaan 95% selisih nilai TWK tes manual dan CAT adalah antara -7.1013 hingga 9.5013. Selang kepercayaan ini memuat nilai 0 didalamnya sehingga dapat disimpulkan bahwa TWK antara manual dan CAT tidak berbeda nyata.

Selang kepercayaan 95% $\mu_1 - \mu_2$ bagi TIU:

$$-0.2429 \leq \mu_2 \leq 14.5429$$

Interpretasi:

dengan tingkat kepercayaan 95% selisih nilai TIU tes manual dan CAT adalah antara -0.2429 hingga 14.5429. Selang kepercayaan ini memuat nilai 0 didalamnya sehingga dapat disimpulkan bahwa TIU antara manual dan CAT tidak berbeda nyata.

*/*selang kepercayaan simultan dan bonferoni*/*

```
data kelompok1;
  set CPNS;
  if metode="Manual";
  variable="twk";  x=twk; output;
  variable="tiu";  x=tiu;  output;
  keep metode variable x;
run;
proc sort;
  by variable;
run;
proc means noprint;
  by variable;
  id metode;
  var x;
  output out=pop1 n=n1 mean=xbar1 var=s21;

data kelompok2;
  set CPNS;
  if metode="cat";
  variable="twk";  x=twk; output;
  variable="tiu";  x=tiu;  output;
  keep metode variable x;
run;
proc sort;
  by variable;
run;

proc means noprint;
  by variable;
  id metode;
  var x;
  output out=pop2 n=n2 mean=xbar2 var=s22;
data combine;
merge pop1 pop2;
by variable;
p=2;
n1=20;
```

```

n2=20;
f=finv(0.95,p,n1+n2-p-1);
t=tinv(1-0.025/p,n1+n2-2);
sp=((n1-1)*s21+(n2-1)*s22)/(n1+n2-2);
losim=xbar1-xbar2-sqrt(p*(n1+n2-2)*f*(1/n1+1/n2)*sp/(n1+n2-p-1));
upsim=xbar1-xbar2+sqrt(p*(n1+n2-2)*f*(1/n1+1/n2)*sp/(n1+n2-p-1));
lobon=xbar1-xbar2-t*sqrt((1/n1+1/n2)*sp);
upbon=xbar1-xbar2+t*sqrt((1/n1+1/n2)*sp);
run;
proc print;
run;

```

Output yang diperoleh adalah sebagai berikut :

```

The SAS System      13:00 Thursday, October 14, 2014    4

Obs   variable      metode   _TYPE_   _FREQ_   n1   xbar1   s21   n2   xbar2   s22
1      tiu           cat       0        20      20    60.10   230.411 20    52.95  154.155
2      twk           cat       0        20      20    65.75   333.776 20    64.55  151.103

Obs   p      f      t      sp      losim   upsim   lobon   upbon
1      2    3.25192  2.33372  192.283  -4.1830  18.4830  -3.0834  17.3834
2      2    3.25192  2.33372  242.439 -11.5256  13.9256 -10.2908  12.6908

```

KASUS RAGAM TIDAK SAMA

a. Hipotesis:

$$H_0: \mu_1 = \mu_2$$

$$H_1: \mu_1 \neq \mu_2$$

Statistik uji:

$$T^2 = (\bar{\mathbf{x}}_1 - \bar{\mathbf{x}}_2)' \left[\left(\frac{s_1}{n_1} + \frac{s_2}{n_2} \right) \right]^{-1} (\bar{\mathbf{x}}_1 - \bar{\mathbf{x}}_2) > c^2$$

Dimana :

$$c^2 = \chi^2_{(\alpha,p)}$$

Matriks kovarians :

$$S_1 = \begin{bmatrix} \text{var}(x_1) & \text{covar}(x_1x_2) \\ \text{covar}(x_2x_1) & \text{var}(x_2) \end{bmatrix} = \begin{bmatrix} 333.776 & 81.868 \\ 81.868 & 230.441 \end{bmatrix}$$

$$S_2 = \begin{bmatrix} \text{var}(x_1) & \text{covar}(x_1x_2) \\ \text{covar}(x_2x_1) & \text{var}(x_2) \end{bmatrix} = \begin{bmatrix} 151.103 & -26.287 \\ -26.287 & 154.155 \end{bmatrix}$$

Sehingga Statistik uji :

$$T^2 = (1.2 \ 6 \ 7.15) \times \begin{pmatrix} 0.0419 & -0.00606 \\ -0.00606 & 0.05288 \end{pmatrix} \times \begin{pmatrix} 1.2 \\ 7.15 \end{pmatrix}$$

$$= 2.6598$$

Nilai Kritis:

$$c^2 = \chi^2_{(\alpha,p)} = 5.9914$$

Keputusan:

Nilai $T^2 < c^2$, maka tidak cukup bukti untuk menolak H_0 .

Interpretasi:

Rata-rata nilai hasil test untuk TWK dan TIU dari peserta yang mengikuti seleksi menggunakan CAT sama dengan peserta yang mengikuti seleksi menggunakan sistem manual (lembar jawaban komputer/LJK).

b. Selang Kepercayaan Simultan

$$a'(\bar{x}_1 - \bar{x}_2) - \sqrt{\chi^2_{p(\alpha)} a' \left(\frac{s_1}{n_1} + \frac{s_2}{n_1} \right) a} \leq a' \mu \leq a'(\bar{x}_1 - \bar{x}_2) + \sqrt{\chi^2_{p(\alpha)} a' \left(\frac{s_1}{n_1} + \frac{s_2}{n_1} \right) a}$$

Selang kepercayaan 95% $\mu_1 - \mu_2$ bagi TWK:

$$-10.852 \leq \mu \leq 13.252$$

Interpretasi:

dengan tingkat kepercayaan 95% selisih nilai TWK tes manual dan CAT adalah antara -10.582 hingga 13.252. Selang kepercayaan ini memuat nilai 0 didalamnya sehingga dapat disimpulkan bahwa TWK antara manual dan CAT tidak berbeda nyata.

Selang kepercayaan 95% $\mu_1 - \mu_2$ bagi TIU:

$$-3.583 \leq \mu_2 \leq 17.883$$

Interpretasi:

dengan tingkat kepercayaan 95% selisih nilai TIU tes manual dan CAT adalah antara -0.2429 hingga 14.5429. Selang kepercayaan ini memuat nilai 0 didalamnya sehingga dapat disimpulkan bahwa TIU antara manual dan CAT tidak berbeda nyata.

DAFTAR PUSTAKA

Johnson RA, Wichern DW. 2007. *Applied Multivariate Statistical Analysis*. Ed ke-6. New Jersey: Prentice Hall, Inc

Sartono, Affendi, et. Al. 2003. Analisis Peubah Ganda. Bogor: IPB Press